2.3 Introduction to EES

Engineering Equation Solver (EES)

- EES solves non-linear sets of equations
  - » iterative solution technique starting from the Guess values
- Equations CAN (but shouldn't usually) be entered in any order
- EES will check that there are enough equations & unknowns
- There are sets of equations for which no solution exists
- Equations are entered in the Equations window
  - » EES is NOT case sensitive
  - » Variable names must start with a letter
  - » 1 equation per line
  - » typical rules (^, *, /, +, -)

Annotating EES Code

- Your equations can be annotated to make the problem more understandable
  - » use comments “”, or {}
  - » comments do not affect the solution

```
"EXAMPLE 3-2: Energy absorption in a lens"
k=1.5 [W/m-K]                               "conductivity"
L=1 [cm]*convert(cm,m)                      "lens thickness"
alpha=0.1 [1/mm]*convert(1/mm,1/m)  "absorption coefficient"
q_flux=0.1 [W/cm^2]*convert(W/cm^2,W/m^2)               "incident energy"
h=20 [W/m^2-K]                              "heat transfer coefficient"
T_a=20 [C]                                  "ambient air temperature"

"Obtain constants of integration"
T_a+q_flux/(k*alpha)-C_2=-q_flux/h-k*C_1/h
-q_flux*exp(-alpha*L)/(k*alpha)+C_1*L+C_2-T_a=-q_flux*exp(-alpha*L)/h-k*C_1/h

"Maximize temperature"
x=x_cm*convert(cm,m)
T=-q_flux*exp(-alpha*x)/(k*alpha)+C_1*x+C_2
```

Using EES Effectively

- It is generally NOT a good idea to enter every equation in a problem and then try to solve it
  - » it is difficult to debug a large problem
  - » it is easy to make small mistakes (e.g., T1 is not T_1)

- Better approach – enter equations systematically and solve your problem often
  - » allows you to quickly isolate and fix problems
  - » reduces frustration

Using EES Effectively

```
"Heat Exchanger Geometry"
D_tube_out=0.0095                                   "tube outer diameter"
t_tube=0.0009                                       "tube thickness"
N_tube_row=10                                       "number of tube rows"
N_tube_col=2                                        "number of tube columns"
H=0.2                                               "height of core"
W=0.204                                             "width of core"
L=0.06                                              "length of core"
t_fin=0.00015                                       "thickness of fins"
p_fin=0.0015                                        "fin pitch - distance between adjacent fins"

"Operating Conditions"
T_f_in=converttemp(C,K,75)                          "inlet temperature of fluid"
T_a_in=converttemp(C,K,20)                          "inlet temperature of air"
P_f=1*convert(atm,Pa)                               "pressure of fluid"
P_a=1*convert(atm,Pa)                               "pressure of air"
Fluid$='Water'                                      "type of fluid"
Vdot_f_gpm=0.5
Vdot_f=Vdot_f_gpm*convert(gal/min,m^3/s)            "volumetric flow rate of fluid"

"Properties"
rho_f=density(Fluid$,T=T_f_in,P=P_f)                "density of fluid"
mu_f=viscosity(Fluid$,T=T_f_in,P=P_f)               "viscosity of fluid"
k_f=conductivity(Fluid$,T=T_f_in,P=P_f)             "conductivity of fluid"
Pr_f=Prandtl(Fluid$,T=T_f_in,P=P_f)                 "Prandtl number of fluid"
c_f=cP(Fluid$,T=T_f_in,P=P_f)                       "specific heat capacity of fluid"
rho_a=density(Air,T=T_a_in,P=P_a)                   "density of air"
mu_a=viscosity(Air,T=T_a_in)                        "viscosity of air"
k_a=conductivity(Air,T=T_a_in)                      "conductivity of air"
Pr_a=Prandtl(Air,T=T_a_in)                          "Prandtl number of air"
c_a=cP(Air,T=T_a_in)                                "specific heat capacity of air"

"Air-Side Performance"
Vdot_a_cfm=60                                       "Guess for the air-side flow rate"
Vdot_a=Vdot_a_cfm*convert(cfm,m^3/s)                "Air side flow rate in SI units"
mdot_a=Vdot_a*rho_a                                 "Mass flow rate on air side"
Nf=W/p_fin                                          "Number of fins on air side"
Ac_a=(H-N_tube_row*D_tube_out)*(W-Nf*t_fin)         "Cross-sectional area available for air flow"
per_a=Nf*2*(H-N_tube_row*D_tube_out)                "Perimeter for flow"
dh_a=4*Ac_a/per_a                                   "Hydraulic diameter of flow"
um_a=Vdot_a/Ac_a                                    "Mean velocity of air"
Re_a=um_a*rho_a*dh_a/mu_a                           "Reynolds number"
f_a=96/Re_a                                         "Friction factor of flow"
DP_a=f_a*(L/dh_a)*rho_a*um_a^2/2                    "Pressure drop"
DP_a_inH2O=DP_a*convert(Pa,inH2O)                   "Pressure drop in inch H2O"
DP_a_inH2O=0.3927-0.003437*Vdot_a_cfm2              "Fan curve"
```

equations are entered in order - therefore, any bottom part of code can be commented out and the rest run

Formatted Equations

- The formatted equations window provides the equations in a more readable format
  - » greek variable names become greek symbols
  - » {} comments do not appear in formatted equations
  - » “” comments do appear in formatted equations
  - » _ gives subscript, _dot gives a dot, others



$$X^2 + Y^3 = 77$$

$$\sqrt{\frac{X}{Y^2 + 1}} = \alpha$$

$$\alpha = 1.23456$$

$$\Delta T = 25$$

$$\dot{m} = X + Y$$

$$V_1 = \alpha^2$$

Parametric Tables

- Parametric tables allow you to run EES multiple times
  - » enables parametric studies
  - » each row in the table corresponds to solving the equations
  - » variables that are set in the table are pasted into the Equations window
  - » the results of the calculation are stored in the other columns
  - » be careful not to over-constrain your problem



Plotting

- Data in any table can be plotted (parametric, lookup, array, or integration)
  - » Select New Plot Window from Plots menu



Plot Formatting

- Plot axes, legends, etc. can be created
- Grid lines
- Data symbols
- Splines
- Adding text or images
- Adding arrows
- Moving plots

Curve Fitting

- Any data that is plotted can be curve fit
- You can use standard curve fits or build a custom functional form



Curve Fitting (continued)



best fit equation can be plotted

details about fit can be examined

best-fit equation can be exported to plot or clipboard

best-fit coefficients are shown

Non-Standard Curve Fitting

(1) select Enter/edit equation



(2) make up your own equation type: z=some function of x where a0, a1, etc are coefficients to be adjusted

Problem #1
(try on your own)

Problem 1

Solve the equation: $a / x + b + c\,x^2 = 1$
for $a = 1$, $b = 2$, and $c = 0.5$.

Make a plot showing how the solution, x, varies as a function of c for values of c ranging from 0.1 to 10. Label your axes. Overlay on your plot the solution if a=2 and a=3 and use a legend to differentiate the curves.

Units

- Most engineering problems involve variables that correspond to physical quantities with units
- EES can:
  - » help you convert between units
  - » allows you to assign a unit to each variable
  - » checks to see if the units assigned to the variables are consistent with the set of equations that you are using
- EES will **NOT** adjust your units so that they are consistent

:

The Convert & Converttemp Functions

The function **Convert('unit1','unit2')** returns the conversion factor to go from 'unit1' to 'unit2'

L=10*convert(ft,inch)

returns the factor 12 inch/ft

$$L = 10 \cdot \left| 12 \cdot \frac{inch}{ft} \right|$$

The function **Converttemp('unit1','unit2',temperature)** returns the temperature converted from 'unit1' to 'unit2'

T=converttemp(C,K,10)

returns the 283 K (which is 10°C)

Units and Engineering Problems

(1) Problem statement (mixed units)

What is the gravitational force exerted on a 2 lbm object placed on the surface of Mercury (where gravity is =3 m/s^2)?

(2) **Convert** all variables to SI units as they are entered:
N, kg, K, m, s, Pa, etc.

and (2) is converted to kg using the **convert** function

(1) the constant 2 has units [lbm]

```
M=2 [lbm]*convert(lbm,kg)      "mass of object"
g=3 [m/s^2]                    "gravity"
```

Units and Engineering Problems

(3) **Set** the units of every variable as it is entered





Units and Engineering Problems

(4) **Solve** the problem in SI units & set the units of each variable in the problem

F=M*g "force on object"



the SI unit of force is N

Units and Engineering Problems

(5) **Check** your equations for dimensional consistency



→ No unit problems were detected.

(6) **Convert** your solution to whatever units you want

F_lbf=F*convert(N,lbf) "force in lbf"

You will never make a unit error and you have ensured that your equations are dimensionally consistent

Problem #2

(try on your own)

Problem 2

A concrete wall is 8 ft high and 10 ft long with a 10 inch thickness. The concrete wall has thermal conductivity 0.7 W/m-K. The wall separates conditioned room air from ambient air. The temperature of the inner surface of the wall is maintained at 25ºC.

Determine the heat flow through the wall ($\dot{q}$) for an outer surface temperature of -5ºC. The equation for this calculation is:

$$\dot{q} = k\,A\frac{\left(T_{in} - T_{out}\right)}{t}$$

where k is the thermal conductivity, A is the area of the wall, t is its thickness, and T_{in} and T_{out} are its inner and outer temperatures.

Make sure you check your units.

Make a plot that shows the heat flow through the wall as a function of wall thickness.

Built-in Functions

- EES provides many built-in functions
- Use the Function Information command (Options menu) to review
- Click the Function Info button for specific help



Property Functions

- EES provides thermodynamic and transport properties for many fluids



Property Functions

- In order to use these functions, you **MUST** set the unit system (Pa, N, K, J, etc.)
  - » indicates the units of the independent properties provided to the functions and
  - » the units of the dependent properties returned from the function





Property Functions

- The property functions are very flexible

Specify any set of 2 independent properties

```
h=enthalpy('R134a',T=300,P=100000)
```

dependent property

fluid

independent properties are temperature & pressure

& their values are 300 **K** and 100000 **Pa**

  » the values could be variables
  » other independent properties are also possible

```
h=enthalpy('R134a',T=300,s=1)
```

enthalpy at T=300 K and s=1 J/kg-K

Property Functions

- ACENTRICFACTOR
- CV and CP
- CONDUCTIVITY
- DENSITY
- DEWPOINT
- ENTHALPY
- ENTROPY
- FUGACITY
- HUMRAT
- INTENERGY
- MOLARMASS
- PRESSURE
- P_CRIT
- PRANDTL

- QUALITY
- RELHUM
- SPECHEAT
- SOUNDSPEED
- SURFACETENSION
- TEMPERATURE
- T_CRIT
- T_TRIPLE
- VISCOSITY
- WETBULB
- VOLUME
- VOLEXPCOE
- V_CRIT

- B= wet-bulb temperature
- D=dew-point temperature
- H=specific enthalpy
- P=pressure
- R=relative humidity
- S=specific entropy
- T=temperature
- U=specific internal energy
- V=specific volume (=1/density)
- W=humidity ratio
- X=quality

Problem #3
(try on your own)

Problem 3
Determine the density of carbon dioxide (in kg/m^3) at a temperature of 500 R and a pressure of 600 bar.

Plot the temperature (in K) of deuterium as a function of pressure from 100 kPa to 10,000 kPa at constant specific volume = 0.1 m^3/kg.

Arrays

- An array is a variable with multiple elements (a vector rather than a scalar)
- You access the elements with brackets []

T[1]=1
T[2]=3
T[3]=4

- Arrays are viewed in the Arrays window



Arrays

- Units can be set for arrays all at once

de-select to collapse array



and set all units for T[] at once



Duplicate Loops

- Duplicate loops carry out a command a certain number of times
- Each time the loop iterates the index increases

index → i; begin value → 1; end value → 10

```
duplicate i=1,10
      T[i]=200
end
```

commands placed between duplicate and end are repeated for each value of i



- Be careful!

```
duplicate i=1,10
      a=10
      T[i]=200
end
```